

9/5

Gestion de fichiers

Nous allons réaliser un programme de gestion de fichiers classique dont les possibilités seront :

- création de structure,
- création de fiche,
- visualisation de l'ensemble des fiches,
- impression de l'ensemble des fiches,
- modification d'une fiche,
- suppression d'une fiche,
- lecture d'un fichier sur disquette,
- sauvegarde d'un fichier sur disquette.

Ce programme fonctionne sur les CPC 664 et 6128, et avec de légères modifications sur CPC 464 qui sont :

1110 PRINT''7) Lire un fichier sur cassette''

1120 PRINT''8) Sauvegarder un fichier sur cassette''

5010 REM Lecture d'un fichier sur cassette

5030 CLS:PRINT''LECTURE D'UN FICHIER SUR CASSETTE'':PRINT

5050 OPENIN ''!'' + N\$

5510 REM Sauvegarde d'un fichier sur cassette

5530 CLS:PRINT''SAUVEGARDE D'UN FICHIER SUR CASSETTE'':PRINT

5550 OPENOUT ''!'' + N\$

La création de structure consiste à définir le nombre d'articles par fiche et le libellé de chaque article.

Une fiche créée sera identifiée par la suite par son numéro de création.

L'ensemble des fiches peut être visualisé sur l'écran, fiche par fiche.

Appuyez sur une touche quelconque pour passer à la fiche suivante.

Pour la sortie sur imprimante, les fiches sont listées les unes derrière les autres, et un message de fin d'impression apparaît à l'écran avant le retour au menu général quand l'impression est finie.

Une fiche peut être modifiée. Son ancien contenu apparaît sur l'écran après avoir donné son numéro d'identification, et le curseur se positionne automatiquement au début de chaque article pour effectuer la modification. Si vous tapez <cr> sur un article, il est annulé.

Une fiche identifiée par son numéro de création peut être supprimée.

Les fichiers lus ou sauvegardés sur disquette ont la structure suivante :

Nombre de libellés

Libellé 1	Libellé n
-----------	-------	-----------

Nombre de fiches

Fiche 1	Fiche n
---------	-------	---------

Le programme écrit en BASIC est le suivant :

```

1000 REM =====
1010 REM          Programme principal
1020 REM =====
1030 CLS:PRINT"GESTION DE FICHIERS"
1040 PRINT:PRINT"Voulez-vous : "
1050 PRINT"1) Creer une structure de fichier"
1060 PRINT"2) Creer une fiche"
1070 PRINT"3) Visualiser l'ensemble des fiches"
1080 PRINT"4) Imprimer l'ensemble des fiches"
1090 PRINT"5) Modifier une fiche"
1100 PRINT"6) Supprimer une fiche"
1110 PRINT"7) Lire un fichier sur disquette"
1120 PRINT"8) sauvegarder un fichier sur disquette"
1130 PRINT"9) Sortir du programme"
1140 PRINT:INPUT "Votre choix ";C
1150 IF C<1 OR C>9 THEN SOUND 1,100,100:GOTO 1000

```

```
1160 ON C GOSUB 2030,2530,3030,3530,4030,4530,5030,5530,6030
1170 IF FIN=0 THEN GOTO 1000 'Boucle sur le menu
1180 END

2000 REM =====
2010 REM          Creation de structure
2020 REM =====
2030 CLS:PRINT"CREATION DE STRUCTURE":PRINT
2040 INPUT"Donnez le nombre d'articles par fiche : ";AF
2050 PRINT:PRINT"Entrez le libelle de chaque article : "
2060 FOR I=1 TO AF
2070   PRINT"Libelle ";I;" : ";:INPUT L$(I)
2080 NEXT I
2090 PRINT:PRINT"La structure est memorisee":FOR I=1 TO 500:NEXT
2100 RETURN

2500 REM =====
2510 REM          Creation d'une fiche
2520 REM =====
2530 CLS:PRINT "CREATION DE FICHE":PRINT
2540 INPUT "Numero de fiche";NF
2550 IF NF>NM THEN NM=NF
2560 FOR I=1 TO AF
2570   PRINT L$(I);:INPUT A$(NF,I)
2580 NEXT I
2590 PRINT:PRINT"Fiche en memoire"
2600 INPUT"Une autre saisie (O/N) : ";A$
2610 A$=UPPER$(A$)
2620 IF A$<>"O" AND A$<>"N" THEN SOUND 1,100,100:GOTO 2600
2630 IF A$="O" THEN 2530
2640 RETURN

3000 REM =====
3010 REM          Visualisation des fiches
3020 REM =====
```

```
3030 CLS:PRINT"VISUALISATION DES FICHES":PRINT
3040 FOR I=1 TO NM
3050   PRINT "Fiche No";I
3060   FOR J=1 TO AF
3070     PRINT L$(J);" : ";A$(I,J)
3080   NEXT J
3090   PRINT:PRINT"Appuyez sur une touche pour visualiser la fiche suivante"
3100   A$=INKEY$:IF A$="" THEN 3100
3110   CLS:PRINT:PRINT
3120 NEXT I
3130 PRINT" Toutes les fiches ont ete visualisees.":FOR I=1 TO 500:NEXT I
3140 RETURN

3500 REM =====
3510 REM           Impression des fiches
3520 REM =====
3530 CLS:PRINT"IMPRESSION DE L'ENSEMBLE DES FICHES":PRINT
3540 FOR I=1 TO NM
3550   PRINT #8,"Fiche No";I
3560   FOR J=1 TO .AF
3570     PRINT #8,L$(J);" : ";A$(I,J)
3580   NEXT J
3590   PRINT #8
3600 NEXT I
3610 PRINT" Toutes les fiches ont ete imprimees.":FOR I=1 TO 500:NEXT I
3620 RETURN

4000 REM =====
4010 REM           Modification d'une fiche
4020 REM =====
4030 CLS:PRINT"MODIFICATION D'UNE FICHE":PRINT
4040 INPUT"Numero de la fiche a modifier ";N
4050 IF N>NM THEN PRINT"Cette fiche n'est pas en memoire":SOUND 1,100,100:FOR I=
1 TO 500:NEXT I:GOTO 4030
```

```
4060 FOR I=1 TO AF
4070 PRINT L$(I);" : ";A$(N,I)
4080 NEXT I
4090 LOCATE 1,4
4100 FOR I=1 TO AF
4110 PRINT L$(I);" ";INPUT A$(N,I)
4120 NEXT I
4130 PRINT:PRINT"Fiche en memoire."
4140 PRINT:INPUT"Une autre modification (O/N)";A$
4150 A$=UPPER$(A$)
4160 IF A$<>"O" AND A$<>"N" THEN SOUND 1,100,100:GOTO 4140
4170 IF A$="O" THEN 4030
4180 RETURN
4500 REM =====
4510 REM          Suppression d'une fiche
4520 REM =====
4530 CLS:PRINT"SUPPRESSION D'UNE FICHE":PRINT
4540 INPUT"Numero de fiche a supprimer";N
4550 IF N>NM THEN SOUND 1,100,100:PRINT"Cette fiche n'est pas en memoire":FOR I=
1 TO 500:NEXT I:GOTO 4500
4560 FOR I=1 TO AF
4570 PRINT L$(I);" : ";A$(N,I)
4580 NEXT I
4590 PRINT:INPUT "Etes-vous sur (O/N) ";A$
4600 A$=UPPER$(A$)
4610 IF A$<>"O" AND A$<>"N" THEN SOUND 1,100,100:GOTO 4530
4620 IF A$="N" THEN 4710
4630 FOR I=1 TO AF
4640 A$(N,I)=" "
4650 NEXT I
4660 PRINT:PRINT"Fiche supprimee"
4670 PRINT:INPUT"Une autre suppression (O/N) ";A$
4680 A$=UPPER$(A$)
```

```
4690 IF A$<>"0" AND A$<>"N" THEN SOUND 1,100,100:GOTO 4530
4700 IF A$="0" THEN 4500
4710 RETURN
5000 REM =====
5010 REM      Lecture d'un fichier sur disquette
5020 REM =====
5030 CLS:PRINT"LECTURE D'UN FICHIER SUR DISQUETTE":PRINT
5040 INPUT "Nom du fichier a lire ";N$
5050 OPENIN N$
5060 INPUT #9,AF 'Nombre de libelles
5070 FOR I=1 TO AF
5080   INPUT #9,L$(I)
5090 NEXT I
5100 INPUT #9,NF 'Nombre de fiches
5110 NM=NF 'Nombre de fiches en memoire
5120 FOR I=1 TO NF
5130   FOR J=1 TO AF
5140     INPUT #9,A$(I,J)
5150   NEXT J
5160 NEXT I
5170 CLOSEIN
5180 PRINT:PRINT"Fichier en memoire":FOR I=1 TO 500:NEXT I
5190 RETURN
5500 REM =====
5510 REM      Sauvegarde d'un fichier sur disquette
5520 REM =====
5530 CLS:PRINT"SAUVEGARDE D'UN FICHIER SUR DISQUETTE":PRINT
5540 INPUT "Nom de la sauvegarde ";N$
5550 OPENOUT N$
5560 PRINT #9,AF 'Nombre de libelles
5570 FOR I=1 TO AF
5580   PRINT #9,L$(I) 'Ecriture des libelles
```

```
5590 NEXT I
5600 PRINT #9,NF 'Nombre de fiches
5610 FOR I=1 TO NF
5620   FOR J=1 TO AF
5630     PRINT #9,A$(I,J) 'Ecriture des fiches
5640   NEXT J
5650 NEXT I
5660 CLOSEOUT
5670 PRINT:PRINT"Fichier sauvegarde sous le nom ";N$:FOR I=1 TO 500:NEXT I
5680 RETURN
6000 REM =====
6010 REM           Sortie du programme
6020 REM =====
6030 CLS:PRINT"SORTIE DU PROGRAMME":PRINT
6040 INPUT"Etes-vous sur (O/N) : ";A$
6050 A$=UPPER$(A$)
6060 IF A$<>"O" AND A$<>"N" THEN SOUND 1,100,100:GOTO 6040
6070 IF A$="O" THEN FIN=1
6080 RETURN
```

Lignes 1000 à 1180 : Programme principal : Affichage du menu et orientation vers un sous-programme spécifique en fonction de l'option choisie.

Lignes 2000 à 2100 : Création d'une structure.

Lignes 2500 à 2640 : Création d'une fiche.

Lignes 3000 à 3140 : Visualisation des fiches.

Lignes 3500 à 3620 : Impression des fiches.

Lignes 4000 à 4180 : Modification d'une fiche.

Lignes 4500 à 4710 : Suppression d'une fiche.

Lignes 5000 à 5190 : Lecture d'un fichier sur disquette.

Lignes 5500 à 5680 : Sauvegarde sur disquette du fichier en mémoire.

Lignes 6000 à 6080 : Fin du programme.

